

"GIVEN: A 4-stroke, 4-cylinder IC engine modeled with the Otto Cycle,"

```
a$ = 'air_ha'
rpm = 2800[rev/min]
P[1] = 14.7[psia]
T[1] = 80[F]
Vol_1 = 34[in^3]*convert(in^3,ft^3)
P[3] = 990[psia]
CR = 10
N_r = 2[rev]  "4-stroke engine"
```

"FIND: (a) the power developed by the engine (hp)  
(b) the thermal efficiency of the cycle  
(c) the mean effective pressure of the engine (hp)"

"SOLUTION:"

"IPROPERTIES"

"State 1: P[1], T[1]"

```
v[1] = volume(a$,P=P[1],T=T[1]);
u[1] = intenergy(a$,P=P[1],T=T[1])
s[1] = entropy(a$,P=P[1],T=T[1])
```

"State 2: v[2], s[2]"

```
CR = v[1]/v[2]
P[2] = pressure(a$,v=v[2],s=s[2])
T[2] = temperature(a$,v=v[2],s=s[2])
u[2] = intenergy(a$,v=v[2],s=s[2])
s[2] = s[1]
```

"State 3: P[3], v[3]"

```
T[3] = temperature(a$,P=P[3],v=v[3])
v[3] = v[2]
u[3] = intenergy(a$,P=P[3],v=v[3])
s[3] = entropy(a$,v=v[3],T=T[3])
```

"State 4: v[4], s[4]=s[3]"

```
P[4] = pressure(a$,v=v[4],s=s[4])
T[4] = temperature(a$,v=v[4],s=s[4])
v[4] = v[1]
u[4] = intenergy(a$,v=v[4],s=s[4])
s[4] = s[3]
```

"THERMODYNAMICS"

"The net work delivered per cylinder is,"

```
W_net = W_34 - W_12
W_34 = m*(u[3] - u[4])
W_12 = m*(u[2] - u[1])
```

"The mass of air trapped in the cylinder is,"

```
m = Vol_1/v[1]
```

"The thermal efficiency of the cycle is,"

```
eta_th = W_net/Q_23
Q_23 = m*(u[3] - u[2])
```

"The mean effective pressure of the cycle is,"

```
mep = W_net*convert(Btu,in-lbf)/(Vol_disp*convert(ft^3,in^3))
Vol_disp = m*(v[1] - v[2])
```

"The power developed by the engine is,"

"!This is a 4-stroke engine, therefore 2 revolutions of the crankshaft are required to execute a cycle. In addition, there are 4 cylinders in the engine. The values calculated above are per cylinder."

```
W_dot_net = 4*W_net*(rpm/N_r)*convert(Btu/min,hp)
```

**Arrays Table: Main**

	$T_i$ [F]	$P_i$ [psia]	$v_i$ [ft <sup>3</sup> /lbm]	$u_i$ [Btu/lbm]	$s_i$ [Btu/lbm-F]
1	80	14.7	13.6	92.03	1.64
2	859.9	363.4	1.36	231.1	1.64
3	3123	990	1.36	717.7	1.851
4	1266	47.07	13.6	311.3	1.851